

```

! Εισαγωγή στοιχείου σε ουρά Q, N θέσεων ΧΩΡΙΣ ΟΛΙΣΘΗΣΗ
AN rear = N ΤΟΤΕ      ! Η ουρά είναι γεμάτη
  ΓΡΑΨΕ 'Η ουρά είναι γεμάτη'
ΑΛΛΙΩΣ                ! Η ουρά δεν είναι γεμάτη
  AN front = 0 ΚΑΙ rear = 0 ΤΟΤΕ      ! Η ουρά είναι άδεια
    front <- 1
    rear <- 1
  ΑΛΛΙΩΣ                ! Η ουρά δεν είναι άδεια, ούτε γεμάτη
    rear <- rear + 1
  ΤΕΛΟΣ_ΑΝ
  ΔΙΑΒΑΣΕ Q[rear]      ! Εισαγωγή του στοιχείου στο πίσω άκρο
ΤΕΛΟΣ_ΑΝ

```

! =====

```

! Εισαγωγή στοιχείου σε ουρά Q, N θέσεων ΜΕ ΟΛΙΣΘΗΣΗ
AN rear = N ΤΟΤΕ
  AN front = 1 ΤΟΤΕ
    ΓΡΑΨΕ 'Η ουρά είναι γεμάτη'
  ΑΛΛΙΩΣ                ! front > 1, άρα υπάρχουν κενές θέσεις μπροστά..
    ! ΟΛΙΣΘΗΣΗ
    k <- 0
    ΓΙΑ i ΑΠΟ front ΜΕΧΡΙ rear ! Εναλλακτικά..
      k <- k + 1                ! ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ (rear - front) + 1
      Q[k] <- Q[i]              ! Q[i] <- Q[front + i - 1]
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ          ! ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
    front <- 1                  ! front <- 1
    rear <- k + 1               ! rear <- i
    ΔΙΑΒΑΣΕ Q[rear]            ! Εισαγωγή του στοιχείου στο πίσω άκρο
  ΤΕΛΟΣ_ΑΝ
ΑΛΛΙΩΣ                ! Η ουρά δεν είναι γεμάτη
  AN front = 0 ΚΑΙ rear = 0 ΤΟΤΕ      ! Η ουρά είναι άδεια
    front <- 1
    rear <- 1
  ΑΛΛΙΩΣ                ! Η ουρά δεν είναι άδεια, ούτε γεμάτη
    rear <- rear + 1
  ΤΕΛΟΣ_ΑΝ
  ΔΙΑΒΑΣΕ Q[rear]            ! Εισαγωγή του στοιχείου στο πίσω άκρο
ΤΕΛΟΣ_ΑΝ

```

! Εξαγωγή στοιχείου απο ουρά Q, ΧΩΡΙΣ ΟΛΙΣΘΗΣΗ

```
AN front = 0 ΚΑΙ rear = 0 ΤΟΤΕ ! Η ουρά είναι άδεια
  ΓΡΑΨΕ 'Η ουρά είναι άδεια'
ΑΛΛΙΩΣ ! Η ουρά δεν είναι άδεια
  ΓΡΑΨΕ Q[front] ! Συνήθως, εκτυπώνουμε το στοιχείο που εξάγεται
  AN front = rear ΤΟΤΕ ! Η ουρά περιείχε ένα στοιχείο, και άδειασε
    front <- 0 ! Οι δείκτες front και rear μηδενίζονται
    rear <- 0 ! Έτσι, αποφεύγουμε να συμβεί front > rear
  ΑΛΛΙΩΣ ! Η ουρά περιέχει περισσότερα απο ένα στοιχεία
    front <- front + 1 ! Εξαγωγή στοιχείου απο το εμπρός άκρο
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΑΝ
```

! =====

! Εξαγωγή στοιχείου απο ουρά Q, ΜΕ ΟΛΙΣΘΗΣΗ κατά 1 θέση εμπρός

```
AN front = 0 ΚΑΙ rear = 0 ΤΟΤΕ ! Η ουρά είναι άδεια
  ΓΡΑΨΕ 'Η ουρά είναι άδεια'
ΑΛΛΙΩΣ ! Η ουρά δεν είναι άδεια
  ΓΡΑΨΕ Q[front] ! Συνήθως, εκτυπώνουμε το στοιχείο που εξάγεται
  AN front = rear ΤΟΤΕ ! Η ουρά περιείχε ένα στοιχείο, και άδειασε
    front <- 0 ! Οι δείκτες front και rear μηδενίζονται
    rear <- 0 ! Έτσι, αποφεύγουμε να συμβεί: front > rear
  ΑΛΛΙΩΣ ! Η ουρά περιέχει περισσότερα απο ένα στοιχεία
    ! ΟΛΙΣΘΗΣΗ κατά 1 θέση εμπρός, μετά απο κάθε εξαγωγή
    ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ rear - 1
      Q[i] <- Q[i+1]
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
    front <- 1
    rear <- rear - 1
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΑΝ
```